<!--#Include File="conn.asp" -->
<!--#Include File="include/inc.asp"-->
<%
'====================================================================
'= Team Elite - Elite Article System
'= Copyright (c) 2005 - 2006 Eason Chan All Rights Reserved.
'=-------------------------------------------------------------------
'= 版权协议：
'=	GPL (The GNU GENERAL PUBLIC LICENSE Version 2, June 1991)
'=-------------------------------------------------------------------
'= 文件名称：Img_List.asp
'= 摘    要：图片文章列表文件
'=-------------------------------------------------------------------
'= 最后更新：eason007
'= 最后日期：2005-05-22
'====================================================================

Dim QueryArray,QueryList
Dim PageNum,PageSize,PageCount,ReCount,Column
Dim i
Dim FieldName(0),FieldValue(0)
Dim RowSize,RowWidth
Dim PageContent

PageNum=EA_Pub.SafeRequest(1,"page",0,1,0)

PageSize=12
RowSize=4
RowWidth=100/RowSize

ReCount=EA_DBO.Get_Article_ImgStat()(0,0)
PageCount=EA_Pub.Stat_Page_Total(PageSize,ReCount)
If PageNum>PageCount And PageCount>0 Then PageNum=PageCount

If PageCount>0 Then 
	QueryArray=EA_DBO.Get_Article_ImgList(PageNum,PageSize)
	
	QueryList="<table border=""0"" align=""center"" cellpadding=""0"" cellspacing=""0"" width=""98%"" style=""word-break: break-all"">"
	QueryList=QueryList&"<tr>"
	For i=0 To UBound(QueryArray,2)
		QueryList=QueryList&"<td width="""&RowWidth&"%"" >"
		QueryList=QueryList&"<table border=""0"" align=""center"" cellpadding=""0"" cellspacing=""0"">"
		QueryList=QueryList&"<tr>"
		QueryList=QueryList&"<td>"
		QueryList=QueryList&"<a href="""&EA_Pub.Cov_ArticlePath(QueryArray(0,i),QueryArray(5,i),QueryArray(6,i),EA_Pub.SysInfo(18))&""" target=""_blank"">"
		'ok
		QueryList=QueryList&"<img src="""&QueryArray(2,i)&""" border=""0"" align=""absmiddle"" onload=""javascript:check(this,90,90)"" onmousemove=""javascript:check(this,90,90)"">"
		QueryList=QueryList&"</a></td>"
		QueryList=QueryList&"</tr>"
		QueryList=QueryList&"<tr>"
		QueryList=QueryList&"<td align=""center"" height=""22"">"
		QueryList=QueryList&"·<a href="""&EA_Pub.Cov_ArticlePath(QueryArray(0,i),QueryArray(5,i),QueryArray(6,i),EA_Pub.SysInfo(18))&""" target=""_blank"">"
		'ok
		QueryList=QueryList&EA_Pub.Add_ArticleColor(QueryArray(3,i),QueryArray(1,i))&"</a>"
		QueryList=QueryList&"</a></td>"
		QueryList=QueryList&"</tr>"
		QueryList=QueryList&"</table>"
		QueryList=QueryList&"</td>"
		
		If (i+1) Mod RowSize=0 Then QueryList=QueryList&"</tr><tr>"
	Next 
		
	If i Mod RowSize=0 Then QueryList=QueryList&"</tr>"
			
	QueryList=QueryList&"<tr>"
	QueryList=QueryList&"<td align=""right"" bgcolor=""efefef"" colspan="""&RowSize&""">"&EA_Temp.PageList(PageCount,PageNum,FieldName,FieldValue)&"</td>"
	QueryList=QueryList&"</tr>"
	QueryList=QueryList&"</table>"
End If

PageContent=EA_Temp.Load_Template(0,"imglist")
	
EA_Temp.Title=EA_Pub.SysInfo(0)&" - 图片文章"
EA_Temp.Nav="<a href=""./""><b>"&EA_Pub.SysInfo(0)&"</b></a> - 图片文章"

PageContent=EA_Temp.Replace_PublicTag(PageContent)
	
PageContent=Replace(PageContent,"{$Query_List$}",QueryList)
	
Response.Write PageContent

Call EA_Pub.Close_Obj
Set EA_Pub=Nothing
%>
